Chapter 9

Sequences and Series

In this chapter, we use Maple to study sequences and series. In the first section, we define sequences using the `seq` command, plot them using the `plot` command and find their limits using the `Limit` and `value` commands. In the second section, we define series using the `Sum` command and compute their sums using the `value` command when the sum can be found exactly. When the sum cannot be found exactly, one must check for convergence and then estimate the sum. The third section shows how Maple's `Limit`, `Int` and `value` commands are useful in performing many convergence tests. When the series converges, it may be approximated by a partial sum of the series. Frequently, an integral can be used to bound the error in this approximation or to improve this approximation as shown in the fourth section. Finally in the fifth section we study Taylor series using the `taylor` and `TaylorApproximation` commands.

9.1 Sequences and Their Limits

To construct a sequence of numbers whose individual terms are given by a formula, first input the formula that describes the terms as a function of n.

```
> a:=n->1/n^2;
```

$$a := n \to \frac{1}{n^2}$$

To have Maple construct the first five terms of this sequence, use the `seq` command. Its first argument is an expression for the terms of the sequence. Its second argument tells Maple which terms to construct.

```
> seq(a(n), n=1..5);
```

$$1, \frac{1}{4}, \frac{1}{9}, \frac{1}{16}, \frac{1}{25}$$

To plot the sequence, you first need a list of points consisting of the index number and the corresponding term. Again, use the `seq` command

```
> pointlist:=[seq([n,a(n)], n=1..5)];
```

$$pointlist := [[1, 1], [2, \frac{1}{4}], [3, \frac{1}{9}], [4, \frac{1}{16}], [5, \frac{1}{25}]]$$

Notice the extra set of square brackets, making this a list of points as required by the `plot` command. (See Section 3.1 and the help on `?list`.) Now plot it:

```
> plot(pointlist, 0..6, style=point, symbol=circle);
```



Often the main task with sequences is to determine the limiting value as n gets large. (This is denoted by $n \to \infty$ which is read as "n goes to infinity.") Maple is very good at calculating the limiting value but sometimes a plot gives a better idea of the sequence's behavior.

EXAMPLE: Examine the behavior of the sequence $b_n = \dfrac{2n-1}{3n+6}$ by plotting the sequence and by finding its limit as $n \to \infty$.

SOLUTION: Define and plot the sequence:

```
> b:=n->(2*n-1)/(3*n+6);
```

$$b := n \to \frac{2\,n-1}{3\,n+6}$$

```
> pointlist:=[seq([n,b(n)], n=1..50)]:
> plot(pointlist, 0..50, 0..1, style=point, symbol=cross);
```



Notice that the `seq` command was terminated with a colon in order to suppress Maple's output. When first trying this, to check for typos, use a semicolon and have Maple display only the first few terms. Then click back on the line and edit it to get all desired terms and use a colon instead. In the `plot` command, if you leave off the option `style=point`, Maple will connect the points with straight lines

Maple's output is a plot of the first 50 terms of the sequence b_n. We see from the plot that as n goes to infinity, the sequence appears to have a limit which is slightly greater than 0.6. However, Maple can find the limit exactly, by using `Limit` and `value`.

```
> Limit(b(n),n=infinity); value(%);
```

$$\lim_{n\to\infty} \frac{2\,n-1}{3\,n+6}$$

$$\frac{2}{3}$$

which is slightly greater than 0.6. Recall that the `Limit` command (with a capital `L`) displays the limit in order to check for typos. The `value` command then evaluates it.

This sequence is so simple that Maple is not needed to compute the limit. (Divide numerator and denominator by n.) However this same procedure can be used to handle more complicated problems.

9.2 Series and Their Sums

A second operation commonly performed on a sequence a_n is to add its terms producing a series $\sum_{n=1}^{\infty} a_n$. (Note: the index of summation, n, does not need to begin with 1 and may end with a finite number or ∞.) To sum the terms, use Maple's `Sum` and `value` commands. If $a_n = \frac{1}{n^2}$, as in the previous section, then a finite sum might be

```
> Sum(a(n),n=5..9); value(%);
```

$$\sum_{n=5}^{9} \frac{1}{n^2}$$

$$\frac{737641}{6350400}$$

while an infinite sum is

```
> Sum(a(n),n=1..infinity); value(%);
```

$$\sum_{n=1}^{\infty} \frac{1}{n^2}$$

$$\frac{\pi^2}{6}$$

Notice that Maple is able to sum this series, while you are not expected to.

Maple may also be able to determine that a series diverges, for example:

```
> c:=n->1/n; Sum(c(n),n=1..infinity); value(%);
```

$$c := n \to \frac{1}{n}$$

$$\sum_{n=1}^{\infty} \frac{1}{n}$$

$$\infty$$

This infinite series is called the harmonic series.

Let's try one more, the geometric series:

```
> a:=n->r^n; Sum(a(n),n=0..infinity); value(%);
```

$$a := n \to r^n$$

$$\sum_{n=0}^{\infty} r^n$$

$$-\frac{1}{r-1}$$

CAUTION: Unfortunately, Maple did not warn us that this series converges to the above value only if $|r| < 1$. For example, you might try to compute the value of the series $\sum_{n=0}^{\infty} 2^n$. So be careful!

Even though Maple sums the above series exactly, most infinite series cannot be summed exactly. So for many series, the goal is to discover whether or not the series converges, and to compute approximate values for its sum when it does converge.

9.3 Convergence of Series

There are four very useful tests for the convergence of a series of positive terms: ratio, root, limit comparison and integral tests. In this section, we will use Maple and these tests to determine whether or not a series of positive terms converges.

Ratio Test: Consider the series $\sum_{n=0}^{\infty} \frac{2^{3n}}{(2n+1)!}$. We input its terms as:

```
> a:=n->2^(3*n)/(2*n+1)!;
```

$$a := n \to \frac{2^{(3\,n)}}{(2\,n+1)!}$$

Again, the terms are entered as functions of n. This allows us to refer back to individual terms when we use the ratio or root test. Here the ratio test is appropriate. We compute:

```
> Limit(a(n+1)/a(n),n=infinity); value(%);
```

$$\lim_{n\to\infty} \frac{2^{(3\,n+3)}\,(2\,n+1)!}{(2\,n+3)!\,2^{(3\,n)}}$$

$$0$$

Since the limit of the ratio of consecutive terms is 0, which is less than 1, the series converges.

Root Test: Similarly, for the root test, consider the series $\displaystyle\sum_{n=0}^{\infty} \frac{3^{2n}}{(2n+1)2^{3n}}$ which we input as:

```
> b:=n->3^(2*n)/(2*n+1)/2^(3*n);
```

$$b := n \to \frac{3^{(2\,n)}}{(2\,n+1)\,2^{(3\,n)}}$$

Using the root test, we compute:

```
> Limit(b(n)^(1/n),n=infinity); value(%);
```

$$\lim_{n\to\infty} \left(\frac{3^{(2\,n)}}{(2\,n+1)\,2^{(3\,n)}}\right)^{(\frac{1}{n})}$$

$$\frac{9}{8}$$

Since the limit is $\frac{9}{8}$ which is greater than 1, the series diverges.

When the limit in the ratio or root test is different from 1, we know whether the series converges or diverges. When this limit is 1, another test is needed. For example, for the series $\displaystyle\sum_{n=2}^{\infty} \frac{\ln(n)}{n^2}$ the limit of the ratios and the limit of the n^{th} roots is 1. So we turn to another test.

Limit Comparison Test: Define the terms of the series $\displaystyle\sum_{n=2}^{\infty} a_n = \sum_{n=2}^{\infty} \frac{\ln(n)}{n^2}$

```
> a:=n->ln(n)/n^2;
```

$$a := n \to \frac{\ln(n)}{n^2}$$

Since $\ln(n)$ grows slower than any positive power of n, we can try a limit comparison test with the series $\displaystyle\sum_{n=2}^{\infty} b_n = \sum_{n=2}^{\infty} \frac{1}{n^{3/2}}$ which is convergent because it is a p-series with $p = \frac{3}{2} > 1$.

```
> b:=n->1/n^(3/2);
```

$$b := n \to \frac{1}{n^{(3/2)}}$$

```
> Limit(a(n)/b(n),n=infinity); value(%);
```

$$\lim_{n\to\infty} \frac{\ln(n)}{\sqrt{n}}$$

$$0$$

This says that the a_n's go to zero faster than the b_n's. So $\sum_{n=2}^{\infty} a_n = \sum_{n=2}^{\infty} \frac{\ln(n)}{n^2}$ also converges.

Integral Test: Again look at the series $\sum_{n=2}^{\infty} \frac{\ln(n)}{n^2}$. Since the terms are positive and decreasing, we can try the integral test.

```
>  a:=n->ln(n)/n^2;
```

$$a := n \to \frac{\ln(n)}{n^2}$$

```
>  Int(a(n),n=2..infinity); value(%);
```

$$\int_2^{\infty} \frac{\ln(n)}{n^2}\, dn$$

$$\frac{1}{2}\ln(2) + \frac{1}{2}$$

Since the integral is finite, the series $\sum_{n=2}^{\infty} \frac{\ln(n)}{n^2}$ converges.

9.4 Error Estimates

One of the more useful aspects of the integral test is its ability to estimate the error in using a partial sum to approximate a series. For example, we saw earlier that the sum of the series $\sum_{n=1}^{\infty} \frac{1}{n^2}$ is $\frac{\pi^2}{6}$. Suppose we didn't know this and added up the first 50 terms of the series. How close are we to the actual sum? Let's see:

```
>  a:=n->1/n^2;
```

$$a := n \to \frac{1}{n^2}$$

```
>  Sum(a(n),n=1..50); S50:=evalf(%);
```

$$\sum_{n=1}^{50} \frac{1}{n^2}$$

$$S50 := 1.625132734$$

```
>  err:=evalf(Pi^2/6-S50);
```

$$err := 0.019801334$$

However, if we did not know the sum is $\frac{\pi^2}{6}$, could we find an upper bound on this error? Yes, using the ideas underlying the integral test! First note that the error is

$$err = \sum_{n=1}^{\infty} \frac{1}{n^2} - \sum_{n=1}^{50} \frac{1}{n^2} = \sum_{n=51}^{\infty} \frac{1}{n^2}$$

This last sum is a lower Riemann sum for $\int_{50}^{\infty} \frac{1}{x^2} dx$ as illustrated below.

```
> rightbox(1/x^2, x=50..75, 25);
```



Consequently,

$$err = \sum_{n=51}^{\infty} \frac{1}{n^2} \leq \int_{50}^{\infty} \frac{dx}{x^2}$$

So we compute:

```
> Int(1/x^2,x=50..infinity); maxerr:=evalf(%);
```

$$\int_{50}^{\infty} \frac{1}{x^2}\, dx$$

$$maxerr := 0.02000000000$$

The error must be less than 0.02. We found it to be .019801334.

Integrals can also be used to improve the estimate. Note that the error

$$err = \sum_{n=51}^{\infty} \frac{1}{n^2}$$

is also an upper Riemann sum for $\int_{51}^{\infty} \frac{1}{x^2} dx$ as illustrated below.

```
> leftbox(1/x^2, x=51..76, 25);
```



So we compute:

```
> Int(1/x^2,x=51..infinity); minerr:=evalf(%);
```

$$\int_{51}^{\infty} \frac{1}{x^2}\, dx$$

$$minerr := 0.01960784314$$

So the error must be at least .01960784314. Putting these facts together, the sum of the series $\sum_{n=1}^{\infty} \frac{1}{n^2}$ must be at least

```
> Smin:=S50+minerr;
```

$$Smin := 1.644740577$$

and at most

```
> Smax:=S50+maxerr;
```

$$Smax := 1.645132734$$

Thus the sum must be in the interval $[S_{min}, S_{max}]$. The midpoint of this interval is a better approximation to the sum:

```
> Save:=(Smin+Smax)/2;
```

$$Save := 1.644936656$$

The error in this new approximation is at most half the width of this interval.

```
> Err:=(Smax-Smin)/2;
```

$$Err := 0.0001960785$$

This is a significantly smaller error. Further, from the above we see that $\frac{\pi^2}{6} \approx 1.644937 \pm 0.000196$. Compare this to Maple's value

```
> evalf(Pi^2/6);
```

$$1.644934068$$

NOTE: This averaging method gives us a much better approximation of the infinite series than that obtained by just using the partial sum of the first 50 terms, as we did earlier. Moreover, the extra work involved is minimal.

9.5 Taylor Polynomials

An extremely useful idea in mathematics is the approximation of complicated functions with simpler ones. In Section 7.4 we studied Fourier cosine expansions which approximated functions by sums of cosines. In this section we study Taylor expansions, which approximate functions by polynomials.

Suppose we want the fifth degree Taylor polynomial of $\sin(x)$ about $x = 0$. The following Maple commands will construct this polynomial.

```
> taylor(sin(x),x=0,6); p:=convert(%,polynom);
```

$$x - \frac{1}{6}x^3 + \frac{1}{120}x^5 + O(x^6)$$

$$p := x - \frac{1}{6}x^3 + \frac{1}{120}x^5$$

A few words about the syntax are in order. Notice that the Maple command `taylor` has three parameters: the expression whose Taylor polynomial we want; the point that we expand about; and an integer. This last parameter—in this case 6—is one more than the degree of the desired Taylor polynomial. In Maple this number refers to the order of the error of the approximating polynomial. In the output, this error is denoted by $O(x^6)$. Notice too that the Maple command `convert(%,polynom)` drops the error term and enables us to assign the polynomial to a variable without having to retype it.

You can graph the function and its Taylor polynomial on the same plot:

```
> plot([sin(x),p],x=-2*Pi..2*Pi,-1.5..1.5);
```



The higher the degree of the Taylor polynomial, the better the approximation.

Further, the `TaylorApproximation` command in the `Student[Calculus1]` package allows you to use a single command to compute a sequence of Taylor polynomials for a given expression:

```
> with(Student[Calculus1]):
> TaylorApproximation(sin(x), x=0, order=1..6);
```

$$x,\ x,\ x - \frac{1}{6}x^3,\ x - \frac{1}{6}x^3,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5,\ x - \frac{1}{6}x^3 + \frac{1}{120}x^5$$

or to plot them together with the original function:

```
> TaylorApproximation(sin(x), x=0, order=1..12, output=plot,
> -2*Pi..2*Pi, view=[-2*Pi..2*Pi,-1.5..1.5],
> functionoptions=[thickness=3], tayloroptions=[thickness=1]);
```



or animate them to see that the Taylor polynomials do approximate the original function.

```
> TaylorApproximation(sin(x), x=0, order=1..12,
> output=animation,-2*Pi..2*Pi, view=[-2*Pi..2*Pi,-1.5..1.5]):
```

Try this yourself, since you cannot see an animation in a book. To animate it, click in the plot and click on the PLAY button which is a triangle on the plot toolbar.

In addition, the Taylor Remainder Formula for the error term can often give a valuable estimate as to how well we have approximated our function. For example, suppose we want to approximate the sine function on the interval $[0, \pi]$ with its Taylor polynomial of degree 7, and we wish to know how good an approximation we have. First we decide to expand about the midpoint of the interval.

```
> taylor(sin(x),x=Pi/2,8):  p:=convert(%,polynom);
```

$$p := 1 - \frac{(x - \frac{\pi}{2})^2}{2} + \frac{(x - \frac{\pi}{2})^4}{24} - \frac{(x - \frac{\pi}{2})^6}{720}$$

The Taylor Remainder Formula says the remainder is

$$E_n = \sin(x) - p = \frac{f^{(n+1)}(c)}{(n+1)!}\left(x - \frac{\pi}{2}\right)^{(n+1)}$$

where c is some number between x and $\pi/2$. In this case, $n = 7$; so we need the eighth derivative of the sine function, evaluated at c. So we compute

```
> diff(sin(x),x$8); subs(x=c,%);
```

$$\sin(x)$$

$$\sin(c)$$

We know the absolute value of sine is never larger than 1. Moreover, the distance between x and $\pi/2$ is always less than or equal to $\pi/2$ (recall that x lies in the interval $[0, \pi]$). We can therefore estimate the error as follows.

$$|E_7| = |\sin(x) - p| = \left| \frac{f^{(8)}(c)}{8!} \left(x - \frac{\pi}{2}\right)^8 \right| < \left| \frac{1}{8!} \left(\frac{\pi}{2}\right)^8 \right|$$

```
> evalf((Pi/2)^8/8!);
```

0.0009192602758

Thus our seventh degree polynomial is uniformly within 0.00092 of the value of the sine function on the interval $[0, \pi]$. To visualize this, we plot the sine function together with the Taylor polynomial.

```
> plot({sin(x),p},x=-Pi/2..3*Pi/2);
```



Notice the agreement is excellent on $[0, \pi]$ and only begins to vary near $-\pi/2$ and $3\pi/2$.

9.6 Summary

We studied sequences using the `seq`, `plot`, `Limit` and `value` commands. Then we studied series using the `Sum`, `value`, `Limit` and `Int` commands. Finally we studied Taylor series using the `taylor`, `convert(%,polynom)` and `TaylorApproximation` commands.

9.7 Exercises

In Exercises 1–7, plot the sequence. Try to determine whether or not the sequence has a limit as $n \to \infty$ and what the limiting value is. Then have Maple compute the limit if it exists.

1. $a_n = \dfrac{3n^3 - 6n^2 + 15}{2n^3 + 18n^2 - 6}$

2. $a_n = \dfrac{3n^3 - 6n^2 + 15}{2n^4 + 18n^2 - 6}$

3. $a_n = \dfrac{2n^4 + 18n^3 - 6}{3n^3 - 6n^2 + 15}$

4. $a_n = \dfrac{2 + (-1)^n n^2}{2n^2 + 3n + 4}$

5. $a_n = \dfrac{\ln(n^2)}{n^{2/3}}$

6. $a_n = \dfrac{n!}{20^n}$

7. $a_n = \left(1 + \dfrac{3}{n}\right)^{1/n}$

In Exercises 8–12 sum the given series using Maple. Then, simultaneously plot the first 10 terms of the sequence of terms a_n (In Maple: `a(n)`.) and the first 10 terms of the sequence of partial sums $S_n = \sum_{k=1}^{n} a_k$. (In Maple: `S:=n->sum(a(k),k=1..n)`.)

8. $\displaystyle\sum_{n=1}^{\infty} \frac{1}{4^n}$

9. $\displaystyle\sum_{n=1}^{\infty} \frac{(-1)^n}{n^2}$

10. $\displaystyle\sum_{n=1}^{\infty} \frac{1}{n^{k+1/2}}$, for $k = -1, 0, 1$

11. $\displaystyle\sum_{n=1}^{\infty} \frac{n^4}{3^n}$

12. $\displaystyle\sum_{n=1}^{\infty} \frac{1}{4 + n^2}$

In Exercises 13 and 14, decide whether the given series converges. In Exercises 15–18, give the values of x for which the series converges. For each of the series, try as many of the convergence tests discussed in this chapter as seem applicable.

13. $\displaystyle\sum_{n=0}^{\infty} \frac{n^3}{2n^5 + n^2 + 2}$

14. $\displaystyle\sum_{n=2}^{\infty} \frac{\ln^2(n)}{n \ln(n^{\ln(n)})}$

15. $\displaystyle\sum_{n=1}^{\infty} \frac{(-1)^n x^n}{n\,3^n}$

16. $\displaystyle\sum_{n=1}^{\infty} \frac{x^n}{e^n}$

17. $\displaystyle\sum_{n=2}^{\infty} \frac{(-1)^n x^{2n}}{n\ln(n)}$

18. $\displaystyle\sum_{n=1}^{\infty} \frac{(2x-3)^n}{n^p}$ for $p = \frac{1}{2}, 1, \frac{3}{2}$ and 2.

19. Find the sixth degree Taylor polynomial of $\cos(2x)$ about $x = \pi/3$. Then use the Taylor Remainder Formula to estimate the error in approximating $\cos(2x)$ with this Taylor polynomial on the interval $[0, \pi]$. Finally, plot both the function and the Taylor polynomial on the same coordinate axes.

20. Find the fourth through the eighth degree Taylor polynomials about $x = 1$ for $\dfrac{x^4 - 15x^2 + 2x - 5}{x^2 - 6}$. Plot all of them and the function on the same graph on the interval $[-1, 2]$. Then animate these plots. Finally use the Taylor Remainder Formula to estimate the error in approximating this function with the eighth degree Taylor polynomial on the interval $[-1, 2]$.

21. Find the 100th degree Taylor polynomial of $x^4 - 2x^2 + 15x - 6$ about $x = 0$ and also about $x = 6$, on the interval $[-2, 8]$. Explain why you should expect this result.

22. Estimate the value of the series $\displaystyle\sum_{n=1}^{\infty} \frac{1}{n^3}$ by using the sum of the first 50 terms of the series and by using the averaging method discussed in Section 9.4. Estimate the error in each approximation.

23. Calculate the seventh and eighth degree Taylor polynomials for the function $\cos(x)$ about $x = \pi/2$. The eighth degree polynomial should be more accurate than the seventh. Is it? How much extra work is involved in evaluating this eighth degree Taylor polynomial than in evaluating the seventh degree one? Next apply the Taylor Remainder Formula to estimate the error in each approximation on the interval $[0, \pi]$. Which error bound should you believe?

24. Compute the 5^{th} degree Taylor polynomial centered at $a = 0$ for the function $f(x) = \sin(x)$. Name this polynomial $p(x)$ and evaluate $p(x^3)$. Now compute the 15^{th} degree Taylor polynomial centered at $a = 0$ for the function $g(x) = \sin(x^3)$. What do you observe?

25. Compute the 6^{th} degree Taylor polynomial centered at $a = 1$ for the function $f(x) = \tan(x)$. Name this polynomial $p(x)$ and evaluate $p(x^4)$. Now compute the 24^{th} degree Taylor polynomial centered at $a = 0$ for the function $g(x) = \tan x^4$. What do you observe?

26. Use the previous 2 problems to make a conjecture concerning the relationship of the n^{th} degree Taylor polynomial centered at $a = 0$ for a function $f(x)$ and the $(k \times n)^{th}$ degree Taylor polynomial centered at $a = 0$ for the function $f(x^k)$.

27. Compute the 5^{th} degree Taylor polynomial centered at $a = 0$ for the function $f(x) = \cos(x)$. Name this polynomial $p(x)$ and evaluate $p(x^3)$. Now compute the 12^{th} degree Taylor polynomial centered at $a = 0$ for the function $g(x) = \cos(x^3)$. What do you observe? In the problems above, we had the same sort of behavior, but the ratio of the Taylor orders would be 3:1 or 15:5. Why do we get the same result here with only a 12:5 ratio?

28. Use an appropriate degree Taylor polynomial centered at $a = 0$ for the function $f(x) = \sin(x^3)$ to approximate

$$\int_0^1 \sin(x^3)\, dx$$

to 15 decimal places of accuracy.

29. Repeat the previous problem for the integral

$$\int_{-3}^{3} e^{-x^2}\, dx$$

30. It is not always clear how to choose the value of the centering point for a Taylor polynomial. Consider the function $f(x) = e^{3x} + 7\sin(x)$ on the interval $[-1, 1]$. Let $p_{3,a}(x)$ denote the generic 3^{rd} degree Taylor polynomial centered at $x = a$ for $f(x)$. Find the value of a so that

$$\int_{-1}^{1} \left[f(x) - p_{3,a}(x)\right]^2\, dx$$

is minimized.